Title

stata.com

graph twoway spike — Twoway spike plots

Syntax

```
twoway spike yvar xvar [if] [in] [, options]
```

options	Description
`vertical`	vertical spike plot; the default
`horizontal`	horizontal spike plot
`base(#)`	value to drop to; default is 0
line_options	change look of spike lines
axis_choice_options	associate plot with alternative axis
twoway_options	titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.

See [G-3] ***line_options***, [G-3] ***axis_choice_options***, and [G-3] ***twoway_options***.

All explicit options are *rightmost*, except `vertical` and `horizontal`, which are *unique*; see [G-4] **concept: repeated options**.

Menu

Graphics > Twoway graph (scatter, line, etc.)

Description

`twoway spike` displays numerical (y,x) data as spikes. `twoway spike` is useful for drawing spike plots of time-series data or other equally spaced data and is useful as a programming tool. For sparse data, also see [G-2] **graph bar**.

Options

`vertical` and `horizontal` specify either a vertical or a horizontal spike plot. `vertical` is the default. If `horizontal` is specified, the values recorded in *yvar* are treated as x values, and the values recorded in *xvar* are treated as y values. That is, to make horizontal plots, do not switch the order of the two variables specified.

In the `vertical` case, spikes are drawn at the specified *xvar* values and extend up or down from 0 according to the corresponding *yvar* values. If 0 is not in the range of the y axis, spikes extend up or down to the x axis.

In the `horizontal` case, spikes are drawn at the specified *xvar* values and extend left or right from 0 according to the corresponding *yvar* values. If 0 is not in the range of the x axis, spikes extend left or right to the y axis.

`base(#)` specifies the value from which the spike should extend. The default is `base(0)`; in the above description of options `vertical` and `horizontal`, this default was assumed.

line_options specify the look of the lines used to draw the spikes, including pattern, width, and color; see [G-3] ***line_options***.

axis_choice_options associate the plot with a particular y or x axis on the graph; see [G-3] ***axis_choice_options***.

twoway_options are a set of common options supported by all `twoway` graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over `by()` groups, and change some advanced settings. See [G-3] ***twoway_options***.

Remarks and examples

stata.com

Remarks are presented under the following headings:

Typical use
Advanced use
Cautions

Typical use

We have daily data recording the values for the S&P 500 in 2001:

```
. use http://www.stata-press.com/data/r13/sp500
(S&P 500)
. list date close change in 1/5
```

	date	close	change
1.	02jan2001	1283.27	.
2.	03jan2001	1347.56	64.29004
3.	04jan2001	1333.34	-14.22009
4.	05jan2001	1298.35	-34.98999
5.	08jan2001	1295.86	-2.48999

The example in [G-2] **graph twoway bar** graphed the first 57 observations of these data by using bars. Here is the same graph presented as spikes:

```
. twoway spike change date in 1/57
```



Spikes are especially useful when there are a lot of data. The graph below shows the data for the entire year:

```
. twoway spike change date
```



Advanced use

The useful thing about `twoway spike` is that it can be combined with other `twoway` plottypes (see [G-2] **graph twoway**):

```
. twoway line close date || spike change date
```

We can improve this graph by typing

```
. twoway
      line close date, yaxis(1)
  ||
      spike change date, yaxis(2)
  ||,
      ysca(axis(1) r(700  1400)) ylab(1000(100)1400, axis(1))
      ysca(axis(2) r(-50 300)) ylab(-50 0 50, axis(2))
             ytick(-50(25)50, axis(2) grid)
      legend(off)
      xtitle("Date")
      title("S&P 500")
      subtitle("January - December 2001")
      note("Source:  Yahoo!Finance and Commodity Systems, Inc.")
      yline(950, axis(1) lstyle(foreground))
```



Concerning our use of

```
yline(950, axis(1) lstyle(foreground))
```

see *Advanced use: Overlaying* in [G-2] **graph twoway bar**.

Cautions

See *Cautions* in [G-2] **graph twoway bar**, which applies equally to `twoway spike`.

Also see

[G-2] **graph twoway bar** — Twoway bar plots

[G-2] **graph twoway dot** — Twoway dot plots

[G-2] **graph twoway dropline** — Twoway dropped-line plots

[G-2] **graph twoway scatter** — Twoway scatterplots